#-------------------------------------------------
#
# Project created by QtCreator 2013-01-08T11:22:26
#
#-------------------------------------------------


# Aufruf:
# ohne KD Chart:
#qmake
# mit KD Chart:
# qmake CONFIG+=kdchart
#optional kdchartinclude=foo und/oder kdchartlib=foo für die Angabe wo KDChart liegt.
QT       += core gui sql

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets

TARGET = qhaushaltsbuch
TEMPLATE = app
OBJECTS_DIR = tmp/obj
MOC_DIR = tmp/moc
UI_HEADERS_DIR = tmp/ui

#für KD Chart
kdchart {
    !isEmpty(kdchartinclude) : INCLUDEPATH += $$kdchartinclude
    !isEmpty(kdchartlib) : LIBS += -L $$kdchartlib
    LIBS+= -lkdchart
    DEFINES+= KDCHART_JA
    KD_CHART="Ja"
    HEADERS += Quellen/Diagramm.h
    SOURCES += Quellen/Diagramm.cpp
} else :  KD_CHART="Nein"

message("KD Chart Unterstützung: "$$KD_CHART)
CODECFORTR = UTF-8
TRANSLATIONS = Sprachen/$$TARGET-en.ts

SOURCES += Quellen/Haupt.cpp\
	Quellen/DlgHaupt.cpp \
    Quellen/DlgKategorie.cpp \
    Quellen/Datenverwaltung.cpp \
    Quellen/Datumeditor.cpp \
    Quellen/Werteditor.cpp

HEADERS  += Quellen/DlgHaupt.h \
    Quellen/Vorgaben.h \
    Quellen/DlgKategorie.h \
    Quellen/Datenverwaltung.h \
    Quellen/Datumeditor.h \
    Quellen/Werteditor.h

FORMS    += Formulare/DlgHaupt.ui \
    Formulare/DlgKategorie.ui
